

TryItOut SPARQL

1. Download **MetEireannStationDetails.ttl** from SPARQL_data zip file from Resources area
2. Import it into your installed triplestore
3. Create queries for the following

Query 1: Get all the triples available

```
SELECT ?s ?p ?o
WHERE {
    ?s ?p ?o .
}
```

Query 2: Count the number of triples

```
SELECT (COUNT(*) as ?Triples)
WHERE {
    ?s ?p ?o .
}
```

Query 3: Retrieve the Height for all the stations

```
PREFIX ex: <http://example.org/>
SELECT ?station ?HeightValue
WHERE {
    ?station ex:Height ?HeightValue .
}
```

Query 3.1: Order by Height.

```
PREFIX ex: <http://example.org/>
SELECT ?station ?HeightValue
WHERE {
    ?station ex:Height ?HeightValue .
}
ORDER BY ASC(?HeightValue)
```

Query 3.3: Get the highest station

```
PREFIX ex: <http://example.org/>
SELECT (MAX(?HeightValue) as ?MaxHeightValue)
WHERE {
    ?station ex:Height ?HeightValue .
}
```

Query 3.4: Return all the Weather station in county Dublin in descending order

```
PREFIX ex: <http://example.org/>
SELECT ?station ?HeightValue ?County
WHERE {
    ?station ex:Height ?HeightValue ;
            ex:hasCountyName ?County .
    FILTER(?County = "Dublin")
}
ORDER BY DESC(?HeightValue)
```

Query 3.6: What are the total height (sum) and average height of all stations in Wicklow?

```
PREFIX ex: <http://example.org/>
SELECT (SUM(?HeightValue) AS ?SumHeight) (AVG(?HeightValue) AS ?AvgHeight)
WHERE {
    ?station ex:Height ?HeightValue ;
            ex:hasCountyName ?County .
    FILTER(?County = "Wicklow")
}
```

Query 4: Find the stations that are still open, which have no ex:closeYear

Option A:

```
PREFIX ex: <http://example.org/>
SELECT ?station ?Name
WHERE {
    ?station ex:hasStationName ?Name.
    FILTER(NOT EXISTS { ?station ex:closeYear ?Year. })
}
```

Option B:

```
PREFIX ex: <http://example.org/>
SELECT ?station ?Name # ?Year #?HeightValue ?County
WHERE {
    ?station ex:hasStationName ?Name.
    OPTIONAL{ ?station ex:closeYear ?Year. }
    FILTER(!bound(?Year))
}
```

Query 5: How many stations were open the year you were born?

```
PREFIX ex: <http://example.org/>
SELECT (COUNT(?station) as ?BirthdayStations)
WHERE {
    ?station ex:hasStationName ?Name;
            ex:startYear ?OYear.
    OPTIONAL{ ?station ex:closeYear ?CYear. }
    FILTER( ?OYear < 1995)
    FILTER( ?CYear >= 1995 || !bound(?CYear))
}
```